Inventory model (Example 1.3 / Leskelä)

Description of the process

A computer store sells laptops each week from Monday to Saturday.
If in the end of Saturday there are fewer than two laptops in stock, a sufficient number of new laptops are ordered so that on Monday morning there will again be five laptops in stock.
(If in the end of the week there are at least two laptops in stock, no new laptops are ordered.)
The demand for laptops during each week is Poisson distributed with mean $\lambda = 3.5$ (i.e., the probability that the random demand D equals $n \in \mathbb{N}$ is $P[D = n] = \frac{1}{n!} \lambda^n e^{-\lambda}$), independently for different weeks.
A customer demanding a laptop buys it at the store if a laptop is available at the time, and otherwise they buy one elsewhere.

State space S = {2, 3, 4, 5}

State $x \in S$ interpreted: "in the beginning of the week, there are x laptops in stock"

Transition matrix

In[1]:=
```
Clear[P, p, λ]
```

Poisson distribution

In[2]:=
```
PDF[PoissonDistribution[λ]][n]
```

Out[2]=

$$\begin{cases} \frac{e^{-\lambda} \lambda^n}{n!} & n \geq 0 \\ 0 & \text{True} \end{cases}$$

Transition probabilities

In[3]:=
```
thresholdToOrderMore = 2;
fullStockCapacity = 5;
p[x_, y_ /; (thresholdToOrderMore ≤ y < fullStockCapacity)] :=
  PDF[PoissonDistribution[λ]][x - y]
p[x_, 5] := (1 - Sum[p[x, y],
     {y, thresholdToOrderMore, fullStockCapacity - thresholdToOrderMore + 1}])
```

Transition matrix

In[7]:=
```
P = Table[Table[p[x, y], {y, thresholdToOrderMore, fullStockCapacity}], {x, 2, 5}];
P // MatrixForm
```

Out[8]//MatrixForm=

$$\begin{pmatrix} e^{-\lambda} & 0 & 0 & 1 - e^{-\lambda} \\ e^{-\lambda}\lambda & e^{-\lambda} & 0 & 1 - e^{-\lambda} - e^{-\lambda}\lambda \\ \frac{1}{2} e^{-\lambda}\lambda^2 & e^{-\lambda}\lambda & e^{-\lambda} & 1 - e^{-\lambda} - e^{-\lambda}\lambda - \frac{1}{2} e^{-\lambda}\lambda^2 \\ \frac{1}{6} e^{-\lambda}\lambda^3 & \frac{1}{2} e^{-\lambda}\lambda^2 & e^{-\lambda}\lambda & 1 - e^{-\lambda}\lambda - \frac{1}{2} e^{-\lambda}\lambda^2 - \frac{1}{6} e^{-\lambda}\lambda^3 \end{pmatrix}$$

Specific demand distribution parameter

In[9]:=
```
λ = 3.5;
P // MatrixForm
```

Out[10]//MatrixForm=

$$\begin{pmatrix} 0.0301974 & 0 & 0 & 0.969803 \\ 0.105691 & 0.0301974 & 0 & 0.864112 \\ 0.184959 & 0.105691 & 0.0301974 & 0.679153 \\ 0.215785 & 0.184959 & 0.105691 & 0.493565 \end{pmatrix}$$

Invariant distribution

In[11]:=
```
Clear[λ]
```

In[12]:=
```
Simplify[
 Solve[Join[Flatten[{{π2, π3, π4, π5}} - {{π2, π3, π4, π5}}.P], {π2 + π3 + π4 + π5}] ==
   {0, 0, 0, 0, 1}, {π2, π3, π4, π5}]]
```

Out[12]=

$$\Big\{\Big\{\pi 2 \to \frac{\left(1+4\,e^{\lambda}+e^{2\lambda}\right)\lambda^3}{-6+6\,e^{3\lambda}+6\,\lambda-3\,\lambda^2+\lambda^3+e^{2\lambda}\left(-18+6\,\lambda+3\,\lambda^2+\lambda^3\right)+2\,e^{\lambda}\left(9-6\,\lambda+2\,\lambda^3\right)},$$

$$\pi 3 \to \frac{3\left(-1+e^{\lambda}\right)\left(1+e^{\lambda}\right)\lambda^2}{-6+6\,e^{3\lambda}+6\,\lambda-3\,\lambda^2+\lambda^3+e^{2\lambda}\left(-18+6\,\lambda+3\,\lambda^2+\lambda^3\right)+2\,e^{\lambda}\left(9-6\,\lambda+2\,\lambda^3\right)},$$

$$\pi 4 \to \frac{6\left(-1+e^{\lambda}\right)^2\lambda}{-6+6\,e^{3\lambda}+6\,\lambda-3\,\lambda^2+\lambda^3+e^{2\lambda}\left(-18+6\,\lambda+3\,\lambda^2+\lambda^3\right)+2\,e^{\lambda}\left(9-6\,\lambda+2\,\lambda^3\right)},$$

$$\pi 5 \to \frac{6\left(-1+e^{\lambda}\right)^3}{-6+6\,e^{3\lambda}+6\,\lambda-3\,\lambda^2+\lambda^3+e^{2\lambda}\left(-18+6\,\lambda+3\,\lambda^2+\lambda^3\right)+2\,e^{\lambda}\left(9-6\,\lambda+2\,\lambda^3\right)}\Big\}\Big\}$$

In[13]:=
```
Π = {{π2, π3, π4, π5}} /. %[[1]];
Π // MatrixForm
```

Out[14]//MatrixForm=

$$\Big(\ \frac{\left(1+4\,e^{\lambda}+e^{2\lambda}\right)\lambda^3}{-6+6\,e^{3\lambda}+6\,\lambda-3\,\lambda^2+\lambda^3+e^{2\lambda}\left(-18+6\,\lambda+3\,\lambda^2+\lambda^3\right)+2\,e^{\lambda}\left(9-6\,\lambda+2\,\lambda^3\right)} \quad \frac{3\left(-1+e^{\lambda}\right)\left(1+e^{\lambda}\right)\lambda^2}{-6+6\,e^{3\lambda}+6\,\lambda-3\,\lambda^2+\lambda^3+e^{2\lambda}\left(-18+6\,\lambda+3\,\lambda^2+\lambda^3\right)+2\,e^{\lambda}\left(9-6\,\lambda+2\,\lambda^3\right)} \quad \frac{}{-6+6\,e^{3\lambda}+6\,\lambda-3\,\lambda^2+\lambda^3+}$$

In[15]:=
```
λ = 3.5;
Π // MatrixForm
```

Out[16]//MatrixForm=

$$\begin{pmatrix} 0.16828 & 0.128473 & 0.0691094 & 0.634137 \end{pmatrix}$$

Limiting distribution

Assume that in the initial state, the stock is full, i . e ., the store has five laptops. (The initial distribution is μ_0 is concentrated on the state with 5 laptops.)

In[17]:=
```
μ0 = {Table[If[x == fullStockCapacity, 1, 0],
     {x, thresholdToOrderMore, fullStockCapacity}]};
μ0 // MatrixForm
```
Out[18]//MatrixForm=
```
( 0  0  0  1 )
```

After t weeks, the distribution is $\mu_t = \mu_0 P^t$. (Theorem 1.5)

In[19]:=
```
μ[t_] := μ0.MatrixPower[P, t]
```

After 4 weeks, the distribution is

In[20]:=
```
μ[4] // MatrixForm
```
Out[20]//MatrixForm=
```
( 0.166279  0.125914  0.0673742  0.640432 )
```

After 52 weeks, the distribution is

In[21]:=
```
μ[52] // MatrixForm
```
Out[21]//MatrixForm=
```
( 0.16828  0.128473  0.0691094  0.634137 )
```

This is already very close to the limiting distribution; the changes from here on are tiny.

In[22]:=
```
μ[780 000 000 000] // MatrixForm
```
Out[22]//MatrixForm=
```
( 0.168279  0.128472  0.069109  0.634133 )
```

Some questions about the model

Q1: What is the average number of computers held in stock on Monday mornings?

Q2: Assume that the store buys its laptops for a wholesale price of 590€, sells them for a retail price of 790€, and has a weekly storage cost of 50€ for each laptop in store in the beginning of the week. What is the average weekly net revenue it makes from its laptop business (including the above revenue components and nothing else)?

Answer to Q1

If we interpret the “average” as the expected value in the limiting (invariant) distribution Π, then Q1 can be easily answered by solving the limiting distribution by the techniques of the previous lecture.

In[23]:=
```
(2 * π2 + 3 * π3 + 4 * π4 + 5 * π5) /. {π2 → 0.16828008288057686`,
  π3 → 0.12847317397462174`, π4 → 0.06910943047589443`, π5 → 0.6341373126689067`}
```
Out[23]=
```
4.1691
```

If we interpret the "average" as the limit as $T \to \infty$ of the time average over the first T weeks, then it is intuitively clear from the convergence to the limiting distribution that the average is given by the calculation above, but a mathematical justification for this is still needed (“ergodicity”).

Answer to Q2

In[24]:=
```
wholesalePrice = 590;
retailPrice = 790;
storageCost = 50;
```

In[27]:=
```
Clear[v, vFun]
```

In[28]:=
```
vFun[x_] := -storageCost * x + (retailPrice - wholesalePrice) *
    (x - Sum[PDF[PoissonDistribution[λ]][k] * (x - k), {k, 0, x}])
```

In[29]:=
```
v = Table[vFun[x], {x, thresholdToOrderMore, fullStockCapacity}];
v // MatrixForm
```

Out[30]//MatrixForm=

$$\begin{pmatrix} 266.783 \\ 352.613 \\ 395.287 \\ 400.198 \end{pmatrix}$$

The expected revenue for the first 10 weeks is (by Theorem 3.2)

In[31]:=
```
Sum[μ0.MatrixPower[P, t].v, {t, 0, 9}]
```

Out[31]=
```
{3735.}
```

The expected average weekly revenue for the first 10 weeks:

In[32]:=
```
% / 10
```

Out[32]=
```
{373.5}
```

Compare with...

..the expected value in the invariant distribution:

In[33]:=
```
Π.v
```

Out[33]=
```
{371.294}
```

By Theorem 3.6 the expected time averages tend to the expected value in the invariant distribution:

In[34]:=
```
Sum[μ0.MatrixPower[P, t].v, {t, 0, 99}] / 100
```

Out[34]=
```
{371.515}
```

In[35]:=
```
Sum[μ0.MatrixPower[P, t].v, {t, 0, 999}] / 1000
```

Out[35]=
```
{371.316}
```

In[36]:=
```
Sum[μ0.MatrixPower[P, t].v, {t, 0, 9999}] / 10 000
```

Out[36]=
```
{371.296}
```